DANIEL F. MERRIAM, Editor

FORTRAN IV PROGRAM FOR CONSTRUCTION OF PI DIAGRAMS WITH THE UNIVAC 1108 COMPUTER

By

JEFFREY WARNER

NASA Manned Spacecraft Center

COMPUTER CONTRIBUTION 33
State Geological Survey
The University of Kansas, Lawrence
1969

in cooperation with the
American Association of Petroleum Geologists
Tulsa, Oklahoma

Editor's Remarks

For a limited time the Geological Survey will make available on magnetic tape the FORTRAN IV program for construction of Pi diagrams for $15.00 (US). For an up-to-date listing of COMPUTER CONTRIBUTIONS write the Editor.



FORTRAN IV PROGRAM FOR CONSTRUCTION OF PI DIAGRAMS WITH THE UNIVAC 1108 COMPUTER

by

Jeffrey Warner

INTRODUCTION

Graphic analysis of fabric data by constructing orientation diagrams is a standard method in structural geology. Turner and Weiss (1963) and Whitten (1966) among others, present methods of constructing the diagrams. This paper describes a computer program, written in FORTRAN IV for the Univac 1108, for constructing the orientation diagrams. The program also has been run on IBM 7040, 7094, 360/40, 360/65, and 360/67 computers.

The program is listed in Appendix II and details of the input deck are given in Appendix I. Samples of input and output are presented in Appendix III.

I wish to thank Drs. C. Klein, D. R. Waldbaum, and M. P. Billings for helpful discussions during writing of the program. Computer funds were provided by Dr. C. Klein through NSF grant 2723 to Harvard University, the Manned Spacecraft Center of NASA, the Maine Geological Survey, Dr. P. Fenner and Dr. P. Butler of the University of Pennsylvania, and the Geology Department of Franklin and Marshall College.

SCOPE OF THE PROGRAM

The program is written to construct different types of diagrams, among which are:

1. Lineation diagrams (orientation of lines)
2. PI diagrams (orientation of poles to planes)
3. Pole orientation diagrams for normal elements determined on the U-stage.
4. Pole orientation diagrams for parallel elements determined on the U-stage.
5. Poles from a set of azimuth and vertical angles.

Input is in the form of field or laboratory measurements as illustrated in Table 1. Input is limited to 500 data points in the present version of the program which uses less than 15K words of core. This limit can be increased to 3000 data points for a 32K machine. In addition, the program has the capability of (1) performing rotations of any data set and constructing a new diagram and (2) combining any number of rotated data sets and constructing a summary diagram.

The amount of rotation either may be read directly as input in the form of a horizontal and vertical angle of rotation or the computer will calculate the rotation angles. In the later instance input is the dip and strike of (1) the plane of input data, and (2) the plane of desired diagram.

Output is of two possible forms; the percent population per 0.01 of the hemisphere may be printed for (1) each input point (point, or short form), and (2) every other print position on the net (about 2500 points; full or long form). All data type and output form options are determined by an input control and title card.

ADVANTAGES AND RUNNING TIME

The chief advantage of using computer-calculated fabric diagrams result from a saving of time. This program will calculate 10 diagrams of 100 poles each, or one diagram of 500 poles in less than one minute. Thus many hours (or even days) of tedious plotting and "counting-out" are eliminated. The hand contouring (including deciding on the contour interval to be used) of a calculated diagram takes less than five minutes.

Another major advantage of calculated diagrams is that their ease of construction encourages one to make multiple analyses of his data. Once the data are punched, they may be used many times in various analyses. For example, several sets of subregions or "domains" may be set up and tested.

Table 1. - Types of input.

MEASUREMENT TYPE	MEASUREMENT	INPUT FORM	
Planes in field	N. 30° E. - 80° N.W.	N* 30 E	80 W
Lineations in field	35° - S. 28° W.	S 28 W	35
U-Stage data or random poles	28° - 31° R.	31	28 R
Azimuth and vertical angle	350° - 20°	350	20

* The "N" in the strike of a plane is optional input.

LOGIC

The chief calculations done by the computer are:

1. Converting the input data into a unit vector representing the point on the sphere.
2. Calculating the angles of rotation, if necessary.
3. Rotating the unit vectors, if necessary.
4. Calculating the projection coordinates of each data point (needed for the point type of count only).
5. Counting the population percent for each one percent area of the hemisphere and filling a matrix representing the projection.
6. Converting the above matrix values into print symbols and putting a circle onto the matrix.
7. Printing and zeroing the matrix.

Step 1, converting the various types into unit vectors is straightforward. The conversion is done in subroutines, one for each input type. The input value of IND on the control-title card determines which subroutine is utilized.

Steps 2, and 3, calculating the rotation angles and performing the rotation is done in SUBROUTINE ROTE. Here again, control is determined by the value of IND.

Step 4, calculating the projection coordinates is straightforward. This is performed in the point type count routine.

Step 5, counting the percent population is performed by one of two procedures, each of which leads to one of the output styles. One procedure, the point form following the method of Lowe and Eitzer (1961) uses each unit vector in turn as the center for a count. The percent matrix thus is filled only where there is an input point. The second procedure, the full form uses the center of every other print position as the count center, in this instance the percent matrix is filled uniformly with no regard to distribution of input points.

Which count procedure is used is determined by the input value of INDEX on the control-title card.

The count itself is made by calculating the angle between the count center and each unit vector (the $\cos^{-1}$ of the dot product). This value is tested against DIFLIM, the central angle that defines the radius of the count circle on the sphere. If the test is passed(that is if DIFLIM is greater than the calculated angle) a count is registered. The percent is then inserted in the appropriate position of the matrix. Note that the counting is done on the sphere, not on the projection. This avoids the problem of distorted areas on projections.

DIFLIM is an input variable on the DUMMY card (card 1). DIFLIM = 0.16 for the count to be a one percent area on the hemisphere.

Step 6, has two parts. The print coordinates of one quarter of the circle are in a data statement. The remainder of the circle is generated by symmetry. The value -999.9 is inserted in these matrix positions in such a manner that data are not overprinted and lost.

A series of IF statements in a loop are used to determine the appropriate print symbol. Print symbols (DUMMY matrix) are read in as input in A1 format as the first data card. As a print symbol is determined for each value in the percent matrix, that value is replaced by the appropriate member of the DUMMY matrix.

In step 7 the percent matrix, now filled with elements of the DUMMY matrix, is printed in A1 format.

LOCAL PROGRAM MODIFICATIONS

Input/Output Devices - Cards PTCT 350-390 of the main program assign various I/0 devices. The cards now are set up for the NASA MSC system. They should be checked to assure compatibility with each local system. The present Logical Unit assignments are:

1. T = 5 Systems input tape (card read).
2. S = 6 Systems output tape (general printing).
3. IPT = -3 Systems punch tape (card punch).
4. ID = 14 A scratch tape.
5. IR = 3 A scratch tape - used in retrieving data when constructing the summary diagrams.

Microfilm Output - The program may be modified to give output on a Stromberg-Carlson 4020 or 4060 microfilm recorder equipped with page type capability. This involves the following source deck additions and changes.

1. Change the following cards to:

```
PTCT 20
C     VERSION 5A  NASA-MSC  UNIVAC 1108

PTCT 220
      DATA (IM(I),I=1,67)/48,1,3*24,13*1,
      6*2,5*3,4*4,3*5,3*6,3*7,3*8,2*9

PTCT 230
      1,2*10,2*11,12,13,14,15,16,2*17,2*18,
      19,20,2*,21,22,23,1/

PTCT 240
      DATA (JM(I),I=1,67)/3*48,1,95,35,36,
      37,38,39,40,41,42,43,44,45,46,4,

PTCT 250
      17,30,31,32,33,34,35,26,27,28,29,30
      25,24,23,2*,22,21,20,19,18,
```

```
PTCT 260
      22*17,16,2*15,14,2*13,12,11,2*10,
      9,8,7,6,5,2*4,2*3,4*2,4*1/

PTCT 750
      DO 330 I=1,47

PTCT 780
      YI=(48-I)*2-48

PTCT 790
      YI=YI*46.0

PTCT 1040
      X=SIN(AZC(I))*RAD*95.0+95.0

PTCT 1050
      Y=COS(AZC(I))*RAD*95.0+95.0

PTCT 1130
      K=48.0-Y*0.25

PTCT 1140
      M=X*0.5+0.5

PTCT 1270
      DO 151,I=6,67

PTCT 1290
      I2=48-I1

PTCT 1390
      DO 812, I=1,47

PTCT 1610
      WRITE (S,110) (TITLE(I),I=1,18)

PTCT 1620
 110  FORMAT (1H1,5X,18A4///50X,5H
      NORTH,3(/52X,1H*))

PTCT 1630
      DO 111 I=1,47
```

2. Add the following cards

```
PTCT 25
C     OUTPUT ON STROMBERG-CARLSON
      MICROFILM UNIT

PTCT 325
C     L IS THE MICROFILM WRITE

PTCT 375
      L=17  (or the appropriate logical unit
      assignment)

PTCT 465
      CALL RSET(1)  (or the appropriate film
      advance)

PTCT 1615
      WRITE (L,110)  (TITLE(I),I=1,18)
```

3. Omit the following cards:

```
PTCT 270
PTCT 280
```

DISCUSSION

The program published by Spencer and Clabaugh (1967) produces the same type of diagrams as the program described here. The most significant difference between the two programs is in the counting routine. In both modes of operation of my program the count is made in a one percent circle on the surface of an imaginary sphere, whereas in Spencer and Clabaugh's program the count is made in a one percent circle on the projection. But, in fact, the projection of a general one percent circle from the surface of the sphere onto the plane of projection is a complex, ellipselike area. An error is introduced which becomes significant near the perimeter of the projection, where there may be a high population.

REFERENCES

Lowe, K.E., and Eitzer, D., 1961, Petrofabric counting by digital computer (abs.): Geol. Soc. America Sp. Paper 68, p. 220.

Spencer, A. B., and Clabaugh, P. S., 1967, Computer program for fabric diagrams: Am. Jour. Sci., v. 265, p. 166-172.

Turner, F. J., and Weiss, L. E., 1963, Structural analysis of metamorphic tectonites: McGraw-Hill Book Company, New York, p. 545.

Whitten, E. H. T., 1966, Structural geology of folded rocks: Rand McNally and Company, Chicago, p. 663.

APPENDIX I Input deck setup

Card I DUMMY and DIFLIM

FORMAT (40A1, 10X, F10.6)

Columns 1-40 DUMMY = the print symbols. This card must be the following:
"b-0123456789ABCDEFGHIJKLMNOP QRSTUVWXYZ*b" °

Columns 53-57 DIFLIM = size of the count circle.
Recommended value for a one percent count = 0.16

Card II Control and Title

FORMAT (I2, I2, I4, 12A6)

Column 2 INDEX = 0 to end job.

1 for the point type of count.

2 for the full type of count.

Column 4 IND = 1 for lineation data.

2 for planes data.

4 for normal elements on the U-stage.

5 for parallel elements on the U-stage.

6 for rotations with dip and strike type input.

7 for rotations with horizontal and vertical angles type input.

8 for azimuth and roe input from cards.

9 for azimuth and roe input from tape IR (this is the summary diagram).

Columns 6-8 N = number of data points in this diagram.

Columns 9-80 TITLE = any alphanumeric string of characters as a title or other identification.

° "b" signifies a blank.

Card III Data

For IND = 1 - one card III for each data point.

FORMAT (3X,A1, F3.0, 1X, A1, F8.0)

Strike: Column 4 ISDR = N or S

Columns 6-7 ST = values of strike in degrees

Column 9 ISD = E or W

Plunge: Columns 16-17 DIP = value of plunge in degrees.

For IND = 2 - one card III for each data point.

FORMAT (4X, F3.0, 1X, A1, F8.0, 1X, A1)

Strike: Column 4 N (optional)

Columns 6-7 ST = value of strike in degrees.

Column 9 ISD = E or W

Dip: Columns 16-17 DIP = value of dip in degrees.

Column 19 IDPD = E or W

For IND = 4 or 5 - one card III for each data point.

FORMAT (2F10.1, A1)

Azimuth: Columns 7-9 ST = value of azimuth in degrees (0-360)

Inclination: Columns 18-19 DIP = value of inclination in degrees.

Column 21 IDPD = R or L for right or left inclination.

For IND = 6 - one card III only.

FORMAT (I1, 1X, 2(F2.0, 1X, A1, F6.0, 1X, A1, 7X))

Column 1 IPP = 0 for each rotated azimuth and roe to be written on tape unit IR. These data are to be read later in this job as data for an IND = 9 (summary) diagram.
1 for each rotated azimuth and roe to be written on tape unit ID. These data are not used later in this job. This tape may

be saved and used in future jobs under IND = 9 control.
3 for each rotated azimuth and roe to be punched on cards. These cards may be input to a future job under IND = 8 control.

Plane of the original data-

Strike: Column 2 N (optional)

Columns 3-4 AST1 - value of strike in degrees

Column 6 IAST1 = E or W

Dip: Columns 11-12 PD1 = value of dip in degrees

Column 14 IPD1 = E or W.

Plane of the desired diagram -

Strike: Column 21 N (optional)

Columns 22-23 AST2 = value of strike in degrees.

Column 25 IAST2 = E or W.

Dip: Columns 30-31 PD2 = value of dip in degrees.

Column 33 IPD2 = E or W.

For IND = 7 - one card III only.

Format (I1, 2F9.2)

Column 1 IPP = 0, 1, or 3 as in IND=6.

Column 6-8 GAMMA = horizontal clockwise component of the rotation in degrees.

Columns 15-17 ALPHA - vertical (up to and to the left) component of the rotation, in degrees.

For IND = 8 - one card III for each data point.

FORMAT (2F10.6)

Columns 1-10 AZC = Azimuth in radians (0 is towards the East increases counter-clockwise).

Columns 11-20 ROEC = central vertical angle in radians (0 to the bottom, pi/2 on the perimeter).

Note - These data cards may be punched during a rotation. (IND = 6 or 7) with IPP = 3.

For IND = 9 - no card III needed.

Data are read from tape IR. Data were written on tape during a rotation (IND = 6 or 7) with IPP = 0 (or with IPP = 1 in a previous job).

For another diagram, with or without a new set of data, repeat cards II and III as necessary.

To end job, use a card II that is blank.

APPENDIX II. Listing of program.

```
C     LISTING OF THE POINT COUNT PROGRAM


C     STERO-NET POINT COUNT AND ROTATION PROGRAM
C     VERSION 5      NASA-MSC   UNIVAC 1108
C     OCT. 24, 1968
C     BY JEFFREY WARNER
C     INPUT ON TITLE AND CONTROL CARD
C     INDEX IN COL 2    = 1 FOR THE POINT TYPE OF PLOT
C                       = 2 FOR THE FULL TYPE OF PLOT
C     IND IN COL 4   = 1 FOR LINEATIONS
C                    = 2 FOR PLANES
C                    = 4 FOR NORMAL ELEMENTS ON THE U-STAGE
C                    = 5 FOR PARALLEL ELEMENTS ON THE U-STAGE
C                    = 6 FOR ROTATIONS WITH TWO DIP AND STRIKES INPUT
C                    = 7 FOR ROTATIONS WITH GAMMA AND ALPHA AS INPUT
C                    = 8 FOR AZIMUTH AND ROE AS INPUT READ FROM CARDS
C                    = 9 FOR AZIMUTH AND ROE INPUT READ FROM TAPE IR
C     N RIGHT JUSTIFIED IN COL 6-8  =  NUMBER OF DATA POINTS
C     TITLE IN COL 9-80 FOR ANY IDENTIFICATION
      DIMENSION TITLE(18),JM(75),IM(75),AZC(500),ROEC(500),DUMMY(40),
     1XV(500),YV(500),ZV(500),CTPC(57,95)
      COMMON AZC,ROEC,T,S,N,IR,ID,IPT,DUMMY,ID17,ID26,ID31,ID35,ID1,ID2
      INTEGER T,S
      DATA (JM(I),I=1,71)/95,1,3*48,47,46,45,44,43,42,41,40,39,38,38,37,
     136,35,34,33,2*32,31,30,29,2*28,27,26,2*25,24,23,2*22,21,20,19,18,
     217,16,15,14,13,12,11,11,10,10,9,8,7,7,6,6,5,5,4,4,3*3,3*2,5*1/
      DATA(IM(I),I=1,71)/3*29,57,11*1,7*2,5*3,4*4,4*5,3*6,7,7,8,8,9,
     12*10,2*11,2*12,2*13,14,15,2*16,2*17,18,19,2*20,21,2*22,23,2*24,
     225,26,27,28/
      DATA (IM(I),JM(I),I=72,74)/3*1,2,2,1/
      DATA ID1,ID2,ID17,ID26,ID31,ID35/1HR,1HL,1HE,1HN,1HS,1HW/
C     ID IS THE SCRATCH TAPE FOR DATA TO BE DISCARDED
C     IPT IS THE GENERAL PUNCH TAPE
C     IR IS THE SCRATCH TAPE FOR DATA TO BE RETREVED
C     S IS THE GENERAL WRITE TAPE
C     T IS THE GENERAL READ TAPE - THE CARD READ
      ID=1
      IPT=-3
      IR=3
      S=6
      T=5
      READ (T,120) (DUMMY(I),I=1,40),DIFLIM
  120 FORMAT (40A1,10X,F10.6)
  105 READ (T,100) INDEX,IND,N,(TITLE(I),I=1,18)
  100 FORMAT (2I2,I4,6A4)
      IF (INDEX.EQ.0) GO TO 101
      WRITE (S,115) (TITLE(I),I=1,18),N
  115 FORMAT (1H1,5X,6A4 ///5X,I4,14H  OBSERVATIONS///)
      IF (IND.EQ.1) CALL POLE1
      IF (IND.EQ.2) CALL POLE2
C     IF (IND.EQ.3) CALL POLE3
      IF (IND.EQ.4) CALL USTAG (IND)
```

```
      IF (IND.EQ.5) CALL USTAG (IND)
      IF (IND.EQ.7) CALL ROTE (IND)
      IF (IND.EQ.6) CALL ROTE (IND)
      IF (IND-8) 173,172,171
  172 READ ( T,160) (AZC(I),ROEC(I),I=1,N)
  160 FORMAT (2F10.6)
      WRITE (S,174)
  174 FORMAT (5X,49HDATA IS AZIMUTH AND ROE FROM PREVIOUS CALCULATION///
     15X,20HDATA READ FROM CARDS///)
      GO TO 173
  171 REWIND IR
      READ (IR,160) (AZC(I),ROEC(I),I=1,N)
      WRITE (S,175)
  175 FORMAT (5X,49HDATA IS AZIMUTH AND ROE FROM PREVIOUS CALCULATION///
     15X,20HDATA READ FROM TAPE ///)
      REWIND IR
  173 DO 106 I=1,N
      ROSIN=SIN(ROEC(I))
      XV(I)=SIN(AZC(I))*ROSIN
      YV(I)=COS(AZC(I))*ROSIN
      ZV(I)=COS(ROEC(I))
  106 CONTINUE
      XN=N
      IF (INDEX.EQ.1)  GO TO 102
      DO 330 I=1,57
C     LONG VERSION OF ROUTINE TO PERFORM POINT COUNT AND FILL THE
C                  CTPC MATRIX WITH THE PERCENT POPULATION PER UNIT AREA
      YI=(58-I)*2-58
      YI=YI*0.01785714
      DO 330 J=1,95,2
      XI=J*2-96
      XI=XI*0.01063829
      COUNT=0.0
      RAD=SQRT(XI*XI+YI*YI)
      IF (RAD.GT.1.01) GO TO 331
      AZ=ATAN2(XI,YI)
      ROE=ATAN(RAD)*2.0
      ZP=COS(ROE)
      ROSN=SIN(ROE)
      XP=SIN(AZ)*ROSN
      YP=COS(AZ)*ROSN
      DO 332 K=1,N
      DOTP=ABS(XV(K)*XP+YV(K)*YP+ZV(K)*ZP)
      ANG=ACOS(DOTP)
      IF (ANG.LE.DIFLIM) COUNT=COUNT+1.0
  332 CONTINUE
  331 CTPC(I,J)=COUNT*100.0/XN
  330 CONTINUE
      GO TO 104
  102 DO 107 I=1,N
C     SHORT VERSION OF ROUTINE TO PERFORM POINT COUNT AND FILL THE CTPC
C                  MATRIX WITH PERCENT POPULATION PER UNIT AREA
      RAD=TAN(ROEC(I)/2.0)
      X=RAD*SIN(AZC(I))*142.5+142.5
      Y=RAD*COS(AZC(I))*142.5+142.5
      COUNT=0.0
      DO 301 J=1,N
      DOTP=ABS(XV(I)*XV(J)+YV(I)*YV(J)+ZV(I)*ZV(J))
      ANG=ACOS(DOTP)
      IF (ANG.LE.DIFLIM) COUNT=COUNT+1.0
  301 CONTINUE
```

```
      COTPC=COUNT*100.0/XN
      K=58.0-Y*0.2
      M=X*0.333333+0.5
      IF (K.LT.1) K=1
      IF (M.LT.1) M=1
      IF (COTPC.GT.CTPC(K,M))CTPC(K,M)=COTPC
  107 CONTINUE
  104 CONTINUE
C     PUT CIRCLE ONTO CTPC MATRIX
      XXX=-999.9
      DO 150 I=1,5
      K=IM(I)
      M=JM(I)
      IF (CTPC(K,M).LE.0.0) CTPC(K,M)=XXX
  150 CONTINUE
      DO 151 I=6,74
      I1=IM(I)
      I2=58-I1
      J1=JM(I)
      J2=96-J1
      IF (CTPC(I1,J1).LE.0.0) CTPC(I1,J1)=XXX
      IF (CTPC(I1,J2).LE.0.0) CTPC(I1,J2)=XXX
      IF (CTPC(I2,J2).LE.0.0) CTPC(I2,J2)=XXX
      IF (CTPC(I2,J1).LE.0.0) CTPC(I2,J1)=XXX
  151 CONTINUE
C     ROUTINE TO CONVERT CTPC MATRIX FROM NUMERICAL VALUES TO PRINT
C                SYMBOLS
      DO 812 I=1,57
      DO 812 J=1,95
      IF (CTPC(I,J).LE.(-999.9)) GO TO 814
      IF (CTPC(I,J).LE.0.0) GO TO 810
      IF (CTPC(I,J).LT.0.5) GO TO 811
      M=3
  813 MM=M-2
      XMM=MM
      IF (CTPC(I,J).LT.XMM) GO TO 815
      M=M+1
      IF (M.LE.37) GO TO 813
      CTPC(I,J)=DUMMY(38)
      GO TO 812
  815 CTPC(I,J)=DUMMY(M)
      GO TO 812
  814 CTPC(I,J)=DUMMY(39)
      GO TO 812
  810 CTPC(I,J)=DUMMY(1)
      GO TO 812
  811 CTPC(I,J)=DUMMY(2)
  812 CONTINUE
C     PRINT AND ZERO MATRIX
      WRITE (S,110)
  110 FORMAT (1H1,51X,1H*)
      DO 111 I=1,57
      WRITE(S,112)(CTPC(I,J),J=1,95)
  112 FORMAT (5X,95A1)
      DO 111 J=1,95
      CTPC(I,J)=0.0
  111 CONTINUE
      CALL PRINT1
      GO TO 105
  101 REWIND ID
      REWIND IR
      STOP
```

```
C     LISTING OF THE LINEATION CONVERSION SUBROUTINE


      SUBROUTINE POLE1
C     ROUTINE FOR LINEATION DATA
      DIMENSION AZC(500),ROEC(500),DUMMY(40)
      COMMON AZC,ROEC,T,S,N,IR,ID,IPT,DUMMY,ID17,ID26,ID31,ID35,ID1,ID2
      INTEGER T,S
      WRITE (S,201)
  201 FORMAT (5X,19HDATA ARE LINEATIONS/1H1,3X,6HSTRIKE,4X,6HPLUNGE///)
      DO 202 I=1,N
      READ  (T,200) ISDR,ST,ISD,DIP
      WRITE (S,200) ISDR,ST,ISD,DIP
  200 FORMAT (3X,A1,F3.0,1X,A1,F8.0)
      ROEC(I)=(90.0-DIP)*0.01745332
      AZ=ST
      IF (ISDR.EQ.ID31) AZ=180.0-ST
      IF (ISD.EQ.ID35) AZ=180.0+ST
      IF (ISD.EQ.ID35.AND.ISDR.EQ.ID26) AZ=360.0-ST
      AZC(I)=AZ*0.01745332
  202 CONTINUE
      RETURN
      END

C     LISTING OF THE PLANES CONVERSION SUBROUTINE


      SUBROUTINE POLE2
C     ROUTINES FOR PLANES DATA
      DIMENSION AZC(500),ROEC(500),DUMMY(40)
      COMMON AZC,ROEC,T,S,N,IR,ID,IPT,DUMMY,ID17,ID26,ID31,ID35,ID1,ID2
      INTEGER T,S
      WRITE (S,220)
  220 FORMAT (5X,15HDATA ARE PLANES/1H1,3X,6HSTRIKE,7X,3HDIP///)
      DO 221 I=1,N
      READ  (T,222) ST,ISD,DIP,IDPD
      WRITE (S,223) ST,ISD,DIP,IDPD
  223 FORMAT (4H   N,F3.0,1X,A1,F8.0,1X,A1)
  222 FORMAT (4X,    F3.0,1X,A1,F8.0,1X,A1)
      ROEC(I)=DIP*0.01745332
      AZ=270.0+ST
      IF (ISD.EQ.ID35) AZ=270.0-ST
      IF (IDPD.EQ.ID35) AZ=90.0-ST
      IF (IDPD.EQ.ID35.AND.ISD.EQ.ID17) AZ=90.0+ST
      AZC(I)=AZ*0.01745332
  221 CONTINUE
      RETURN
```

```
C       LISTING OF THE U-STAGE CONVERSIONS SUBROUTINE


        SUBROUTINE USTAG (IND)
C       ROUTINES FOR U-STAGE DATA
        DIMENSION AZC(500),ROEC(500),DUMMY(40)
        COMMON AZC,ROEC,T,S,N,IR,ID,IPT,DUMMY,ID17,ID26,ID31,ID35,ID1,ID2
        INTEGER T,S
        IF (IND.EQ.4) WRITE (S,241)
        IF (IND.EQ.5) WRITE (S,240)
  240   FORMAT (5X,16HDATA ARE U-STAGE///5X,38HLINEAR ELEMENTS PARALLEL TO
       1 SCOPE AXIS/1H1,5X,7HAZIMUTH,4X,11HINCLINATION///)
  241   FORMAT (5X,16HDATA ARE U-STAGE///5X,36HLINEAR ELEMENTS NORMAL TO S
       1COPE AXIS/  1H1,5X,7HAZIMUTH,4X,11HINCLINATION///)
        DO 243 I=1,N
        READ (T,246) ST,DIP,IDPD
  246   FORMAT (2F10.1,A1)
        WRITE (S,242) ST,DIP,IDPD
  242   FORMAT (2F10.0,1X,A1)
        IF (ST.GT.180.0) ST=ST-180.0
        AZ=ST+90.0
        IF (IDPD.EQ.ID1) AZ=ST-90.0
        IF (ST.LT.90.0) GO TO 245
        AZ=ST-90.0
        IF (IDPD.EQ.ID1) AZ=ST+90.0
  245   RO=90.0-DIP
        IF (IND.EQ.4) GO TO 244
        RO =DIP
        AZ=AZ+180.0
  244   IF (AZ.LT.0.0) AZ=360.0+AZ
        IF (AZ.GT.360.0) AZ=AZ-360.0
        ROEC(I)=RO *0.01745332
        AZC(I)=AZ*0.01745332
  243   CONTINUE
        RETURN
```

```
C     LISTING OF THE ROTATIONS SUBROUTINE


      SUBROUTINE ROTE (IND)
C       ROUTINE TO ROTATE DATA  CONTAINS TWO PARTS
      DIMENSION AZC(500),ROEC(500),DUMMY(40),XV(500),YV(500),ZV(500)
      COMMON AZC,ROEC,T,S,N,IR,ID,IPT,DUMMY,ID17,ID26,ID31,ID35,ID1,ID2
      INTEGER T,S
C     PART ONE      READ DATA AS DIP AND STRIKE OF ORIGINAL AND NEW
C                PLANES OF DATA
C     CALCULATE ROTATION AS GAMMA AND ALPHA
      IF(IND.EQ.6)READ(T,402)IPP,AST1,IAST1,PD1,IPD1,AST2,IAST2,PD2,IPD2
  402 FORMAT (I1,1X,2(F2.0,1X,A1,F6.0,1X,A1,7X))
      IF (IND.EQ.7) READ (T,403) IPP,GAMMA,ALPHA
  403 FORMAT (I1,2F9.2)
C     IPP = 0 IF AZC AND ROEC ARE TO BE USED AGAIN IN THIS RUN
C     IPP = 1 IF AZC AND ROEC ARE NOT NEEDED AGAIN
C     IPP = 3 IF AZC AND ROEC ARE TO BE PUNCHED
      IF (IPP-1) 308,309,310
  308 IP=IR
      WRITE (S,305)
  305 FORMAT (5X,27HROTATION OF EXISTING POINTS///5X,
     127HDATA ON TAPE TO BE RETREVED///)
      GO TO 311
  309 IP=ID
      WRITE (S,307)
  307 FORMAT (5X,27HROTATION OF EXISTING POINTS///5X,
     122HDATA WILL BE DISCARDED///)
      GO TO 311
  310 IP=IPT
      WRITE (S,306)
  306 FORMAT (5X,27HROTATION OF EXISTING POINTS///5X,20HDATA WILL BE PUN
     1CHED///)
  311 IF (IND.EQ.7) GO TO 404
      WRITE (S,4021) AST1,IAST1,PD1,IPD1,AST2,IAST2,PD2,IPD2
 4021 FORMAT (1H0/5X,37HDIP AND STRIKE OF ORIGINAL DATA PLANE//10X,1HN,
     1F4.0,A1,F7.0,A1////5X,39HDIP AND STRIKE OF ROTATED TO DATA PLANE//
     210X,1HN,F4.0,A1,F7.0,A1///)
C     AZ(1 AND 2) =0.0 TO EAST
      AZ1=180.0+AST1
      AZ2=180.0+AST2
      IF (IAST1.EQ.ID17) AZ1=360.0-AST1
      IF (IAST2.EQ.ID17) AZ2=360.0-AST2
      IF (IAST1.EQ.ID17.AND.IPD1.EQ.ID17) AZ1=180.0-AST1
      IF (IAST2.EQ.ID17.AND.IPD2.EQ.ID17) AZ2=180.0-AST2
      IF(IAST1.EQ.ID35.AND.IPD1.EQ.ID35) AZ1=AST1
      IF(IAST2.EQ.ID35.AND.IPD2.EQ.ID35) AZ2=AST2
      VERT2=PD2*0.01745332
      VERT1=PD1*0.01745332
      AZ2=(AZ2-AZ1)*0.01745332
      ALPHA=VERT1
      XYC=SIN(VERT2)
      ZZC=COS(VERT2)
      XXC=XYC*COS(AZ2)
      YYC=XYC*SIN(AZ2)
      IF (ABS(YYC).GT.0.99) GO TO 4025
      XZC=SQRT(XXC*XXC+ZZC*ZZC)
      ROT=ATAN2(ZZC,XXC)+ALPHA
      IF (ROT.GT.3.14159) ROT=ROT-3.14159
```

```
      IF (ROT.LT.0.0) ROT=ROT+3.14159
      ZZC=SIN(ROT)*XZC
      XXC=XZC*COS(ROT)
      XYC=SQRT(XXC*XXC+YYC*YYC)
      IF (ZZC.GT.0.99)  GO TO 4026
 4025 AZ2=ATAN2(YYC,XXC)+AZ1*0.01745332
      GO TO 4027
 4026 AZ2=0.0
 4027 VERT2=ATAN2(XYC,ZZC)
      GAMMA=AZ2*57.2957
      ALPHA=VERT2*57.2957
 404  IF (GAMMA.LT.0.01.AND.ALPHA.LT.0.01) GO TO 320
      ALPHR=ALPHA*0.01745332
      GAMR =GAMMA*0.01745332
      WRITE (S,4022) GAMMA,ALPHA
 4022 FORMAT (10H0  GAMMA =,F10.2///9H  ALPHA =,F10.2///)
C     PART TWO     ROTATING THE DATA
      DO 300 I=1,N
      AZ=AZC(I)+GAMR
      XY=SIN(ROEC(I))
      ZZ=COS(ROEC(I))
      XX=XY*SIN(AZ)
      YV(I)=XY*COS(AZ)
      IF (ABS(YV(I)).GT.0.99) GO TO 301
      XZ=SQRT(XX*XX+ZZ*ZZ)
      AROT=ATAN2(ZZ,XX) +ALPHR
      IF (AROT.GT.3.14159) AROT=AROT-3.14159
      IF (AROT.LT.0.0) AROT=AROT+3.14159
      ZV(I)=XZ*SIN(AROT)
      XV(I)=XZ*COS(AROT)
      XY=SQRT(XV(I)*XV(I)+YV(I)*YV(I))
      IF (ZV(I).GT.0.99) GO TO 302
 301  AZC(I)=ATAN2(XV(I),YV(I))-GAMR
      ROEC(I)=ATAN2(XY,ZV(I))
      GO TO 303
 302  AZC(I)=0.0
      ROEC(I)=0.0
 303  WRITE (IP,304) AZC(I),ROEC(I)
 304  FORMAT (2F10.6)
 300  CONTINUE
 320  CONTINUE
      RETURN
```

```
C     LISTING OF THE SYMBOL TABLE PRINT SUBROUTINE


      SUBROUTINE PRINT1
C     ROUTINE TO PRINT SYMBOL TABLE
      DIMENSION AZC(500),ROEC(500),DUMMY(40)
      COMMON AZC,ROEC,T,S,N,IR,ID,IPT,DUMMY,ID17,ID26,ID31,ID35,ID1,ID2
      INTEGER T,S
      WRITE (S,550) (DUMMY(I),I=1,4)
  550 FORMAT (28H1  SYMBOLS ON STERO-NET MEAN////5X,6HSYMBOL, 7X,13HPERC
     1ENT RANGE//8X,A1,9X,9H0 OF LESS/8X,A1,9X,1H0,3X,3H1/2/8X,A1,8X,
     23H1/2,4X,A1)
      DO 551 I=4,37
      MBO=I-3
      MTO=I-2
  551 WRITE (S,552) DUMMY(I),MBO,MTO
  552 FORMAT (8X,A1,8X,I2,4X,I2)
      WRITE (S,553) DUMMY(38)
  553 FORMAT (8X,A1,8X,14H35 AND GREATER///////)
      RETURN
      END
```

APPENDIX III. Input/output examples.

```
* * * * IN-PUT EXAMPLE 1 * * * *


 0-123456789ABCDEFGHIJKLMNOPQRSTUVWXYZ*              0.159
 1 1  10      LINEATION TEST  *  *  EXAMPLE
   N 20 E       85
   N 45 W       20
   S 45 E       35
   N 50 E       60
   N 65 E       70
   N 65 E       70
   N 70 E       60
   N 10 E       70
   N 15 E       75
   N 30 W       60
 1 7  10        ROTATION TEST FOR ALPHA AND GAMMA  *  *  EXAMPLE
1   15.0    90.0
 2 2  10   PLANES TEST  *  *  EXAMPLE
   N 25 E       90 E
   N 20 E       85 E
   N 45 W       20 E
   N 50 E       60 W
   N 65 E       70 W
   N 65 E       70 W
   N 70 E       60 W
   N 10 E       70 W
   N 15 E       75 E
   N 30 W       60 E
 1 4  10         U-STAGE, NORMAL ELEMENTS TEST  *  *  EXAMPLE
      135        50 L
      135        50 L
      135        50 L
      135        50 L
      200.       80 R
      200.       80 R
      200.       80 R
       30        30 R
       30        30 R
       30        30 L
 1 6  10     ROTATED DIAGRAM OF THE PREVIOUS PLOT
0N15 E     15 W      N 30 E    80 W
 2 5  10        U-STAGE PARALLEL ELEMENTS TEST  *  *  EXAMPLE
      135        50 L
      135        50 L
      135        50 L
      135        50 L
      200.       80 R
      200.       80 R
      200.       80 R
       30        30 R
       30        30 R
       30        30 L
 2 6  10     ROTATED DIAGRAM OF THE PREVIOUS PLOT
0N15 E     15 W      N 30 E    80 W
 2 9  20     SUMMARY DIAGRAM OF THE TWO ROTATED U-STAGE PLOTS - FULL TYPE PLOT
 1 9  20     SUMMARY DIAGRAM OF THE TWO ROTATED U-STAGE PLOTS - POINT TYPE PLOT
```

```
* * * * IN-PUT EXAMPLE 2 * * * *

     ***** ONLY THE CONTROL - TITLE CARD FOLLOWS *****


 1 2 162        BEDS OF SECTION 23, BUCKFIELD QUADRANGLE

 2 2 162        BEDS OF SECTION 23, BUCKFIELD QUADRANGLE
```

* * * * OUT-PUT EXAMPLE 1 * * * *

***** PARTS OF THE FOLLOWING HAVE BEEN COMPRESSED AND OMITTED *****

* * * * OUT-PUT EXAMPLE 2 * * * *

******* ONLY THE PLOTS FOLLOW *****

LINEATION TEST * * EXAMPLE

10 OBSERVATIONS

DATA ARE LINEATIONS

STRIKE	PLUNGE
N20. E	85.
N45. W	20.
S45. E	35.
N50. E	60.
N65. E	70.
N65. E	70.
N70. E	60.
N10. E	70.
N15. E	75.
N30. W	60.

SYMBOLS ON STERO-NET MEAN

SYMBOL	PERCENT RANGE	
	0 OF LESS	
0	0	1/2
-	1/2	1
1	1	2
2	2	3
3	3	4
4	4	5
5	5	6
6	6	7
7	7	8
8	8	9
9	9	10
A	10	11
B	11	12
C	12	13
D	13	14
E	14	15
F	15	16
G	16	17
H	17	18
I	18	19
J	19	20
K	20	21
L	21	22
M	22	23
N	23	24
O	24	25
P	25	26
Q	26	27
R	27	28
S	28	29
T	29	30
U	30	31
V	31	32
W	32	33
X	33	34
Y	34	35
Z	35 AND GREATER	


A
A
K
A
K
A
K
A
A

ROTATION TEST FOR ALPHA AND GAMMA * * EXAMPLE

10 OBSERVATIONS

ROTATION OF EXISTING POINTS GAMMA = 15.00

DATA WILL BE DISCARDED ALPHA = 90.00

SYMBOLS ON STERO-NET MEAN

SYMBOL	PERCENT RANGE	
	0 OF LESS	
0	0	1/2
-	1/2	1
1	1	2
2	2	3
3	3	4
4	4	5
5	5	6
6	6	7
7	7	8
8	8	9
9	9	10
A	10	11
B	11	12
C	12	13
D	13	14
E	14	15
F	15	16
G	16	17
H	17	18
I	18	19
J	19	20
K	20	21
L	21	22
M	22	23
N	23	24
O	24	25
P	25	26
Q	26	27
R	27	28
S	28	29
T	29	30
U	30	31
V	31	32
W	32	33
X	33	34
Y	34	35
Z	35 AND GREATER	


K
A
K
A
K
A
A
A
A

PLANES TEST * * EXAMPLE

10 OBSERVATIONS

DATA ARE PLANES

STRIKE	DIP
N25. E	90. E
N20. E	85. E
N45. W	20. E
N50. E	60. W
N65. E	70. W
N65. E	70. W
N70. E	60. W
N10. E	70. W
N15. E	75. E
N30. W	60. E

SYMBOLS ON STERO-NET MEAN

SYMBOL	PERCENT RANGE	
	0 OF LESS	
0	0	1/2
-	1/2	1
1	1	2
2	2	3
3	3	4
4	4	5
5	5	6
6	6	7
7	7	8
8	8	9
9	9	10
A	10	11
B	11	12
C	12	13
D	13	14
E	14	15
F	15	16
G	16	17
H	17	18
I	18	19
J	19	20
K	20	21
L	21	22
M	22	23
N	23	24
O	24	25
P	25	26
Q	26	27
R	27	28
S	28	29
T	29	30
U	30	31
V	31	32
W	32	33
X	33	34
Y	34	35
Z	35 AND GREATER	


A
K
U

U-STAGE, NORMAL ELEMENTS TEST * * EXAMPLE

10 OBSERVATIONS

DATA ARE U-STAGE

LINEAR ELEMENTS NORMAL TO SCOPE AXIS

AZIMUTH	INCLINATION
135.	50. L
135.	50. L
135.	50. L
135.	50. L
200.	80. R
200.	80. R
200.	80. R
30.	30. R
30.	30. R
30.	30. L

SYMBOLS ON STERO-NET MEAN

SYMBOL	PERCENT RANGE	
	0 OF LESS	
0	0	1/2
-	1/2	1
1	1	2
2	2	3
3	3	4
4	4	5
5	5	6
6	6	7
7	7	8
8	8	9
9	9	10
A	10	11
B	11	12
C	12	13
D	13	14
E	14	15
F	15	16
G	16	17
H	17	18
I	18	19
J	19	20
K	20	21
L	21	22
M	22	23
N	23	24
O	24	25
P	25	26
Q	26	27
R	27	28
S	28	29
T	29	30
U	30	31
V	31	32
W	32	33
X	33	34
Y	34	35
Z	35 AND GREATER	


K
Z
U
A

ROTATED DIAGRAM OF THE PREVIOUS PLOT

10 OBSERVATIONS

ROTATION OF EXISTING POINTS

DATA ON TAPE TO BE RETREVED

DIP AND STRIKE OF ORIGINAL DATA PLANE

N 15.E   15.W

DIP AND STRIKE OF ROTATED TO DATA PLANE

N 30.E   80.W

GAMMA =   328.74

ALPHA =   65.55


Z
U
A*
K

U-STAGE PARALLEL ELEMENTS TEST * * EXAMPLE

10 OBSERVATIONS

DATA ARE U-STAGE

LINEAR ELEMENTS PARALLEL TO SCOPE AXIS

SYMBOLS ON STERO-NET MEAN

AZIMUTH	INCLINATION
135.	50. L
135.	50. L
135.	50. L
135.	50. L
200.	80. R
200.	80. R
200.	80. R
30.	30. R
30.	30. R
30.	30. L

SYMBOL	PERCENT RANGE	
	0 OF LESS	
0	0	1/2
-	1/2	1
1	1	2
2	2	3
3	3	4
4	4	5
5	5	6
6	6	7
7	7	8
8	8	9
9	9	10
A	10	11
B	11	12
C	12	13
D	13	14
E	14	15
F	15	16
G	16	17
H	17	18
I	18	19
J	19	20
K	20	21
L	21	22
M	22	23
N	23	24
O	24	25
P	25	26
Q	26	27
R	27	28
S	28	29
T	29	30
U	30	31
V	31	32
W	32	33
X	33	34
Y	34	35
Z	35 AND GREATER	



ROTATED DIAGRAM OF THE PREVIOUS PLOT

10 OBSERVATIONS

ROTATION OF EXISTING POINTS

DATA ON TAPE TO BE RETREVED

DIP AND STRIKE OF ORIGINAL DATA PLANE

N 15.E 15.W

DIP AND STRIKE OF ROTATED TO DATA PLANE

N 30.E 80.W

GAMMA = 328.74

ALPHA = 65.55



SUMMARY DIAGRAM OF THE TWO ROTATED U-STAGE PLOTS - FULL TYPE PLOT

20 OBSERVATIONS

DATA IS AZIMUTH AND ROE FROM PREVIOUS CALCULATION

DATA READ FROM TAPE

SYMBOLS ON STERO-NET MEAN

SYMBOL	PERCENT RANGE	
	0 OF LESS	
0	0	1/2
-	1/2	1
1	1	2
2	2	3
3	3	4
4	4	5
5	5	6
6	6	7
7	7	8
8	8	9
9	9	10
A	10	11
B	11	12
C	12	13
D	13	14
E	14	15
F	15	16
G	16	17
H	17	18
I	18	19
J	19	20
K	20	21
L	21	22
M	22	23
N	23	24
O	24	25
P	25	26
Q	26	27
R	27	28
S	28	29
T	29	30
U	30	31
V	31	32
W	32	33
X	33	34
Y	34	35
Z	35 AND GREATER	



SUMMARY DIAGRAM OF THE TWO ROTATED U-STAGE PLOTS - POINT TYPE PLOT

20 OBSERVATIONS

DATA IS AZIMUTH AND ROE FROM PREVIOUS CALCULATION

DATA READ FROM TAPE

SYMBOLS ON STERO-NET MEAN

SYMBOL	PERCENT RANGE	
	0 OF LESS	
0	0	1/2
-	1/2	1
1	1	2
2	2	3
3	3	4
4	4	5
5	5	6
6	6	7
7	7	8
8	8	9
9	9	10
A	10	11
B	11	12
C	12	13
D	13	14
E	14	15
F	15	16
G	16	17
H	17	18
I	18	19
J	19	20
K	20	21
L	21	22
M	22	23
N	23	24
O	24	25
P	25	26
Q	26	27
R	27	28
S	28	29
T	29	30
U	30	31
V	31	32
W	32	33
X	33	34
Y	34	35
Z	35 AND GREATER	


K
F
A
F
5*
K
A
5

BEDS OF SECTION 23, BUCKFIELD QUADRANGLE

162 OBSERVATIONS

DATA ARE PLANES

STRIKE	DIP	STRIKE	DIP		
N20. E	80. W	N30. E	75. W	N30. W	75. W
N15. E	70. W	N25. E	65. W	N50. W	70. W
N10. E	70. W	N30. E	45. W	N45. W	90. E
N20. E	80. E	N40. E	55. W	N30. E	50. W
N40. E	75. W	N10. E	80. W	N40. W	70. E
N20. E	70. W	N 5. E	70. W	N30. E	60. W
N20. E	65. W	N15. E	75. W	N25. E	65. W
N10. E	70. W	N 0. E	75. W	N40. E	70. W
N45. E	70. W	N10. E	65. W	N20. W	90. E
N80. E	50. E	N40. W	60. W	N15. E	75. W
N80. E	60. W	N35. W	70. W	N15. W	45. E
N60. W	70. E	N 0. E	75. W	N20. W	80. W
N25. E	90. E	N10. E	75. W	N35. W	60. E
N 5. E	80. E	N20. E	70. W	N40. W	70. W
N35. E	70. W	N25. W	90. E	N15. W	75. W
N20. E	75. E	N20. W	70. E	N55. W	80. E
N25. E	80. E	N 5. E	90. E	N20. W	75. E
N20. E	75. E	N50. W	50. W	N10. W	80. E
N 5. E	65. E	N20. W	90. E	N30. W	80. W
N30. E	75. W	N20. W	90. E	N15. W	60. E
N35. W	85. W	N15. W	75. E	N20. E	80. E
N15. E	70. W	N30. W	85. W	N 0. E	65. E
N70. W	90. E	N20. W	90. E	N 5. W	70. E
N 0. E	75. W	N10. W	70. W	N10. E	90. E
N 5. E	80. W	N15. W	75. W	N20. W	80. E
N 5. W	75. W	N20. E	75. W	N20. E	30. E
N40. E	70. W	N30. E	60. W	N20. W	60. E
N50. E	70. W	N35. E	70. E	N10. W	70. E
N40. E	65. W	N30. E	80. W	N35. W	70. E
N30. E	85. E	N35. E	70. W	N10. W	90. E
N20. E	80. W	N45. E	65. W	N10. W	80. W
N20. E	75. W	N 0. E	90. E	N20. W	70. W
N10. W	75. E	N45. E	50. W	N20. W	75. W
N15. E	85. W	N15. E	65. W	N35. W	80. E
N15. E	60. W	N 5. E	90. E	N10. W	75. W
N15. E	70. W	N 5. W	90. E	N15. E	60. W
N 0. E	80. W	N15. W	60. W	N15. E	70. W
N45. W	85. W	N35. E	70. W	N10. E	80. W
N 0. E	80. W	N15. E	70. W	N10. E	75. W
N 0. E	70. W	N 0. E	70. W	N30. E	55. W
N 5. E	70. W	N10. E	65. W	N20. E	55. W
N15. W	80. E	N 0. E	80. W	N25. W	60. E
N 0. E	85. W	N15. W	70. W	N70. E	50. W
N10. E	75. W	N25. W	70. E	N30. W	70. W
N10. W	80. W	N30. E	75. W	N20. W	90. E
N 5. E	70. W	N20. W	30. E	N10. E	50. W
N30. E	70. W	N 5. W	80. W	N40. W	40. E
N 0. E	60. W	N60. W	70. E	N25. E	75. W
N 5. W	75. E	N 0. E	75. E	N10. E	55. W
N 5. E	75. W	N45. W	70. E	N 5. W	80. W
N 5. W	65. W	N30. E	80. W	N 0. E	50. W
N25. W	60. W	N10. E	60. W	N 5. W	70. W
N 5. E	60. W	N60. E	40. W	N10. W	90. E
N30. E	60. W	N89. E	40. W	N10. W	80. W





KANSAS GEOLOGICAL SURVEY COMPUTER PROGRAM
THE UNIVERSITY OF KANSAS, LAWRENCE

PROGRAM ABSTRACT

Title (If subroutine state in title):

STERO-NET POINT COUNT AND ROTATION PROGRAM

Date: 15 November 1968

Author, organization: Jeffrey Warner, Geology and Geochemistry Branch, NASA Manned Spacecraft Center, Houston, Texas 77058

Direct inquiries to: Jeffrey Warner, Mail Code-TH2 (Author)

Name: Jeffrey Warner

Address: TH2 NASA Manned Spacecraft Center
Houston, Texas 77058

Purpose/description: Construction, rotation, and summation of Pi diagrams for structural geology and petro fabrics.

Mathematical method: Convert data to unit vectors; dot product of vectors is counted; count inserted in matrix; count values converted to symbols; matrix printed.

Restrictions, range: 500 data points per diagram.

Computer manufacturer: Univac

Model: 1108

Programming language: FORTRAN IV

Memory required: 15 K

Approximate running time: 5 sec/diagram

Special peripheral equipment required: None

Remarks (special compilers or operating systems, required word lengths, number of successful runs, other machine versions, additional information useful for operation or modification of program) hundreds of successful runs of different versions for IBM 7040, 7094, 360/65, and 360/67 computers. To correct for computer errors involving significant figure in calculating dot products add the following statements

```
If (DOTP. GE. 1.0. and. DOTP. LT. 1.001) DOTP=0.099          PTCT 935
If (DOTP. GE. 1.0. and. DOTP. LT. 1.001) DOTP=0.099          PTCT 1085
```

COMPUTER CONTRIBUTIONS

1. Mathematical simulation of marine sedimentation with IBM 7090/7094 computers, by J.W.Harbaugh, 1966.
2. A generalized two-dimensional regression procedure, by J.R.Dempsey, 1966.
3. FORTRAN IV and MAP program for computation and plotting of trend surfaces for degrees 1 through 6, by Mont O'Leary, R.H.Lippert, and O.T.Spitz, 1966.
4. FORTRAN II program for multivariate discriminant analysis using an IBM 1620 computer, by J.C.Davis and R.J.Sampson, 1966.
5. FORTRAN IV program using double Fourier series for surface fitting of irregularly spaced data, by W.R. James, 1966.
6. FORTRAN IV program for estimation of cladistic relationships using the IBM 7040, by R.L.Bartcher, 1966.
7. Computer applications in the earth sciences: Colloquium on classification procedures, edited by D.F. Merrian, 1966.
8. Prediction of the performance of a solution gas drive reservoir by Muskat's Equation, by Apolonio Baca, 1967.
9. FORTRAN IV program for mathematical simulation of marine sedimentation with IBM 7040 or 7094 computers, by J.W.Harbaugh and W.J.Wahlstedt, 1967.
10. Three-dimensional response surface program in FORTRAN II for the IBM 1620 computer, by R.J.Sampson and J.C.Davis, 1967.
11. FORTRAN IV program for vector trend analyses of directional data, by W.T.Fox, 1967.
12. Computer applications in the earth sciences: Colloquium on trend analysis, edited by D.F.Merriam and N.C.Cocke, 1967.
13. FORTRAN IV computer programs for Markov chain experiments in geology, by W.C.Krumbein, 1967.
14. FORTRAN IV programs to determine surface roughness in topography for the CDC 3400 computer, by R.D. Hobson, 1967.
15. FORTRAN II program for progressive linear fit of surfaces on a quadratic base using an IBM 1620 computer, by A.J.Cole, C.Jordan, and D.F.Merriam, 1967.
16. FORTRAN IV program for the GE 625 to compute the power spectrum of geological surfaces, by J.E.Esler and F.W.Preston, 1967.
17. FORTRAN IV program for Q-mode cluster analysis of nonquantitative data using IBM 7090/7094 computers, by G.F.Bonham-Carter, 1967.
18. Computer applications in the earth sciences: Colloquium on time-series analysis, D.F.Merriam, editor, 1967.
19. FORTRAN II time-trend package for the IBM 1620 computer, by J.C.Davis and R.J.Sampson, 1967.
20. Computer programs for multivariate analysis in geology, D.F.Merriam, editor, 1968.
21. FORTRAN IV program for computation and display of principal components, by W.J. Wahlstedt and J.C. Davis, 1968.
22. Computer applications in the earth sciences: Colloquium on simulation, D.F.Merriam and N.C.Cocke, editors, 1968.
23. Computer programs for automatic contouring, by D.B.McIntyre, D.D.Pollard, and R.Smith, 1968.
24. Mathematical model and FORTRAN IV program for computer simulation of deltaic sedimentation, by G.F. Bonham-Carter and A.J.Sutherland, 1968.
25. FORTRAN IV CDC 6400 computer program for analysis of subsurface fold geometry, by E.H.T.Whitten, 1968.
26. FORTRAN IV computer program for simulation of transgression and regression with continuous-time Markov models, by W.C.Krumbein, 1968.
27. Stepwise regression and nonpolynomial models in trend analysis, by A.T.Miesch and J.J.Connor, 1968.
28. KWIKR8 a FORTRAN IV program for multiple regression and geologic trend analysis, by J.E.Esler, P.F. Smith, and J.C.Davis, 1968.
29. FORTRAN IV program for harmonic trend analysis using double Fourier series and regularly gridded data for the GE 625 computer, by J.W.Harbaugh and M.J.Sackin, 1968.
30. Sampling a geological population (workshop on experiment in sampling), by J.C.Griffiths and C.W. Ondrick, 1968.
31. Multivariate procedures and FORTRAN IV program for evaluation and improvement of classifications, by Ferruh Demirmen, 1969.
32. FORTRAN IV programs for canonical correlation and canonical trend-surface analysis, by P.J.Lee, 1969.
33. FORTRAN IV program for construction of Pi diagrams with the Univac 1108, by Jeffrey Warner, 1969.